# TB 13-19 Pulsatile Blood Flow 4-3-18

## N. T. Gladd

**Initialization:** Be sure the file *NTGUtilityFunctions.m* is in the same directory as that from which this notebook was loaded. Then execute the cell immediately below by mousing left on the cell bar to the right of that cell and then typing “shift” + “enter”. Respond “Yes” in response to the query to evaluate initialization cells.

```
SetDirectory[NotebookDirectory[]];
(* set directory where source files are located *)
Get["NTGUtilityFunctions.m"]; (* Load utilities package *)
```

## Purpose

This is the 12th in a series of notebooks in which I work through material and exercises in the magisterial new book *Modern Classical Physics* by Kip S. Thorne and Roger D. Blandford. If you are a physicist of any ilk, BUY THIS BOOK. You will learn from a close reading and from solving the exercises.

**Exercise 13.19** *Example: Pulsatile Blood Flow*
Consider the pulsatile flow of blood through one of the body's larger arteries. The pressure gradient $dP/dz = P'(t)$ consists of a steady term plus a term that is periodic, with the period of the heart's beat.

(a) Assuming laminar flow with $\mathbf{v}$ pointing in the $z$ direction and being a function of radius and time, $\mathbf{v} = v(\varpi, t)\mathbf{e}_z$, show that the Navier-Stokes equation reduces to $\partial v/\partial t = -P'/\rho + \nu\nabla^2 v$.

(b) Explain why $v(\varpi, t)$ is the sum of a steady term produced by the steady (time-independent) part of $P'$, plus terms at angular frequencies $\omega_0, 2\omega_0, \ldots$, produced by parts of $P'$ that have these frequencies. Here $\omega_0 \equiv 2\pi/(\text{heart's beat period})$.

(c) Focus on the component with angular frequency $\omega = n\omega_0$ for some integer $n$. For what range of $\omega$ do you expect the $\varpi$ dependence of $v$ to be approximately Poiseuille [Eq. (13.80)], and what $\varpi$ dependence do you expect in the opposite extreme, and why?

(d) By solving the Navier-Stokes equation for the frequency-$\omega$ component, which is driven by the pressure-gradient term $dP/dz = \Re(P'_\omega e^{-i\omega t})$, and by imposing appropriate boundary conditions at $\varpi = 0$ and $\varpi = a$, show that

$$v = \Re\left[\frac{P'_\omega e^{-i\omega t}}{i\omega\rho}\left(1 - \frac{J_0(\sqrt{i}\, W\varpi/a)}{J_0(\sqrt{i}\, W)}\right)\right]. \qquad (13.83)$$

Here $\Re$ means take the real part, $a$ is the artery's radius, $J_0$ is the Bessel function, $i$ is $\sqrt{-1}$, and $W \equiv \sqrt{\omega a^2/\nu}$ is called the (dimensionless) *Womersley number.*

(e) Plot the pieces of this $v(\varpi)$ that are in phase and out of phase with the driving pressure gradient. Compare with the prediction you made in part (b). Explain the phasing physically. Notice that in the extreme non-Poiseuille regime, there is a boundary layer attached to the artery's wall, with sharply changing flow velocity. What is its thickness in terms of $a$ and the Womersley number? We study boundary layers like this one in Sec. 14.4 and especially Ex. 14.18.

I will solve this problem in the context of a model for blood flow through an artery.

Consider Navier-Stokes equation with gravity assumed negligible

$$\frac{\partial \mathbf{v}}{\partial t} + (\mathbf{v}\cdot\nabla)\cdot\mathbf{v} = -\frac{\nabla P}{\rho} + \frac{1}{\rho}\frac{\partial \sigma_{ik}}{\partial x_k} \qquad (1)$$

For incompressible flow, this can be written

$$\frac{\partial \mathbf{v}}{\partial t} + (\mathbf{v}\cdot\nabla)\cdot\mathbf{v} = -\frac{\nabla P}{\rho} + \alpha\nabla^2\mathbf{v} \qquad (2)$$

where I use $\alpha = \eta/\rho$ to represent the kinetic viscosity. The Mathematica font for “nu” looks too much like “vee”.

For laminar flow in the z-direction, v = v(r) $\mathbb{1}_z$

$$\frac{\partial v_z(r,\ t)}{\partial t} = -\frac{1}{\rho}\frac{\partial P(z,\ t)}{\partial z} + v\nabla^2 v_z(r,\ t) \qquad (3)$$

because $(\mathbf{v}\cdot\nabla)\cdot\mathbf{v} = v_z(r)\,\partial/\partial z\, v_z(r) = 0$

This is the equation called for in part (a) of TB exercise (13.19)

In general, the time-dependence of pressure gradient term that drives the flow is modeled by a Fourier series with $\omega_0$ being the frequency of a heart beat.

$$\frac{1}{\rho}\frac{\partial P(r,\ t)}{\partial z} \equiv \mathcal{P}\sum_{k=0}^{n} \epsilon_n e^{i\,k\,\omega_0\,t} \tag{4}$$

The k = 0 term corresponds to a time-independent pressure driver. This is the explanation sought in part (b) of exercise TB13-19.

Note

```
Laplacian[f[R], {R, θ, z}, "Cylindrical"]
```

$$\frac{f'[R]}{R} + f''[R]$$

So the pde to be solved is

$$\frac{\partial\, v_z(r,\ t)}{\partial t} = -\mathcal{P}\sum_{k=0}^{n} \epsilon_n e^{i\,n\,\omega\,t} + \alpha\left(\frac{1}{r}\frac{\partial v_z(r,\ t)}{\partial r} + \frac{\partial^2 v_z(r,\ t)}{\partial r^2}\right) \tag{5}$$

```
w[1] = D[vz[r, t], t] == -𝒫 Inactivate[Sum][ϵk Exp[I k ω0 t], {k, 0, n}] +
   ν (D[vz[r, t], r]/r + D[vz[r, t], {r, 2}])
```

$$vz^{(0,1)}[r,\ t] == -\mathcal{P}\sum_{k=0}^{n} e^{i\,k\,t\,\omega_0}\,\epsilon_k + \nu\left(\frac{vz^{(1,0)}[r,\ t]}{r} + vz^{(2,0)}[r,\ t]\right)$$

Like all good physicists, I seek a dimensionless form

```
w[2] = w[1] /. vz → Function[{r, t}, v0 V[r / a, t / t0]]
```

$$\frac{v0\,V^{(0,1)}\left[\frac{r}{a},\ \frac{t}{t0}\right]}{t0} == -\mathcal{P}\sum_{k=0}^{n} e^{i\,k\,t\,\omega_0}\,\epsilon_k + \nu\left(\frac{v0\,V^{(1,0)}\left[\frac{r}{a},\ \frac{t}{t0}\right]}{a\,r} + \frac{v0\,V^{(2,0)}\left[\frac{r}{a},\ \frac{t}{t0}\right]}{a^2}\right)$$

The natural scale length is the radius of of an artery, a

```
w[3] = w[2] /. r → a R /. t → T t0
```

$$\frac{v0\,V^{(0,1)}[R,\ T]}{t0} == -\mathcal{P}\sum_{k=0}^{n} e^{i\,k\,T\,t0\,\omega_0}\,\epsilon_k + \nu\left(\frac{v0\,V^{(1,0)}[R,\ T]}{a^2\,R} + \frac{v0\,V^{(2,0)}[R,\ T]}{a^2}\right)$$

```
w[4] = MapEqn[(# t0 / v0) &, w[3]] // ExpandAll
```

$$V^{(0,1)}[R,\ T] == -\frac{t0\,\mathcal{P}\sum_{k=0}^{n} e^{i\,k\,T\,t0\,\omega_0}\,\epsilon_k}{v0} + \frac{t0\,\nu\,V^{(1,0)}[R,\ T]}{a^2\,R} + \frac{t0\,\nu\,V^{(2,0)}[R,\ T]}{a^2}$$

Choose the the time scale to be the diffusion time across the arterial radius

$$\text{def[t0]} = \frac{t0\,\nu}{a^2} == 1$$

$$\frac{t0\,\nu}{a^2} == 1$$

```
w[5] = w[4] /. Sol[def[t0], ν]
```

$$V^{(0,1)}[R, T] == -\frac{t0\,\mathcal{P}\sum_{k=0}^{n} e^{i\,k\,T\,t0\,\omega_0}\,\epsilon_k}{v0} + \frac{V^{(1,0)}[R, T]}{R} + V^{(2,0)}[R, T]$$

Define a dimensionless flow speed

$$\text{def[v0]} = \frac{t0\,\mathcal{P}}{v0} == 1$$

$$\frac{t0\,\mathcal{P}}{v0} == 1$$

```
w[6] = w[5] /. Sol[def[v0], v0]
```

$$V^{(0,1)}[R, T] == -\sum_{k=0}^{n} e^{i\,k\,T\,t0\,\omega_0}\,\epsilon_k + \frac{V^{(1,0)}[R, T]}{R} + V^{(2,0)}[R, T]$$

The characteristic dimensionless oscillation frequency is defined

$$\text{def}[\Omega] = \Omega == \omega_0\,t0$$

$$\Omega == t0\,\omega_0$$

where $\Omega = 1/(2\,\pi)$ is a canonical value corresponding to a pulse rate of 60/minute.

```
w[7] = w[6] /. Sol[def[Ω], ω0]
```

$$V^{(0,1)}[R, T] == -\sum_{k=0}^{n} e^{i\,k\,T\,\Omega}\,\epsilon_k + \frac{V^{(1,0)}[R, T]}{R} + V^{(2,0)}[R, T]$$

This constitutes the starting equation for the analysis

Seek a solution having the form

$$V(R,\ T)\ = V_0(R,\ t) + \sum_{k=0}^{n} \epsilon_k\, V_k(R)\, e^{i\,k\,\Omega\,T}$$

Note that the k = 0 term corresponds to a constant driver

```
w[8] = w[7] /. Sum[e^(i k T Ω) ∈_k, {k, 0, n}] → 1 /. V → Function[{R, T}, V_0[R]]
```

$$0 == -1 + \frac{V_0'[R]}{R} + V_0''[R]$$

The solution of this equation corresponds to Poiseuille flow.

```
w[9] = DSolve[{w[8], V_0[1] == 0, V_0'[0] == 0}, V_0[R], R][[1, 1]]
```

$$V_0[R] \to \frac{1}{4}\left(-1 + R^2\right)$$

For future use, I define the function

```
Clear[VPoiseuille];
VPoiseuille[R_] := 1/4 (-1 + R^2)
```

The general oscillatory problem is solved by separation of variables, and the particular ansatz

```
w[10] = w[7] /. Sum[e^(i k T Ω) ∈_k, {k, 0, n}] → e^(i k T Ω) ∈_k /. V → Function[{R, T}, ∈_k V_k[R] Exp[I k Ω T]]
```

$$i\, e^{i k T \Omega} k\, \Omega\, \epsilon_k V_k[R] == -e^{i k T \Omega} \epsilon_k + \frac{e^{i k T \Omega} \epsilon_k V_k'[R]}{R} + e^{i k T \Omega} \epsilon_k V_k''[R]$$

```
w[11] = MapEqn[(Simplify[#/(∈_k Exp[I k T Ω])]) &, w[10]]
```

$$i\, k\, \Omega\, V_k[R] == -1 + \frac{V_k'[R]}{R} + V_k''[R]$$

The solution is

```
w[12] = DSolve[w[11], V_k[R], R][[1, 1]]
```

$$V_k[R] \to \frac{i}{k\,\Omega} + \text{BesselJ}\left[0, (-1)^{3/4} \sqrt{k}\, R \sqrt{\Omega}\right] C[1] + \text{BesselY}\left[0, -(-1)^{3/4} \sqrt{k}\, R \sqrt{\Omega}\right] C[2]$$

The condition that the solution be well behaved at R = 0 requires C[2] = 0

```
w[13] = w[12] /. C[2] → 0
```

$$V_k[R] \to \frac{i}{k\,\Omega} + \text{BesselJ}\left[0, (-1)^{3/4} \sqrt{k}\, R \sqrt{\Omega}\right] C[1]$$

The boundary condition is $V_k$(R =1 ) = 0

```
w[14] = (w[13] // RE) /. R → 1 /. V_k[1] → 0
```

$$0 == \frac{i}{k\,\Omega} + \text{BesselJ}\left[0,\ (-1)^{3/4}\sqrt{k}\ \sqrt{\Omega}\right]\text{C}[1]$$

Thus

```
Sol[w[14], C[1]]
```

$$\text{C}[1] \to -\frac{i}{k\,\Omega\,\text{BesselJ}\left[0,\ (-1)^{3/4}\sqrt{k}\ \sqrt{\Omega}\right]}$$

```
w[15] = w[13] /. Sol[w[14], C[1]] // Simplify
```

$$V_k[R] \to \frac{i\left(1 - \frac{\text{BesselJ}\left[0,\ (-1)^{3/4}\sqrt{k}\ R\sqrt{\Omega}\right]}{\text{BesselJ}\left[0,\ (-1)^{3/4}\sqrt{k}\ \sqrt{\Omega}\right]}\right)}{k\,\Omega}$$

With less elucidation, the original ode could have been solved immediately

```
w[16] = DSolve[{w[11], V_k[1] == 0, V_k'[0] == 0}, V_k[R], R][[1, 1]]
```

$$V_k[R] \to \left(i\left(\text{BesselJ}\left[0,\ (-1)^{3/4}\sqrt{k}\ \sqrt{\Omega}\right] - \text{BesselJ}\left[0,\ (-1)^{3/4}\sqrt{k}\ R\sqrt{\Omega}\right]\right)\right) \Big/ \left(k\,\Omega\,\text{BesselJ}\left[0,\ (-1)^{3/4}\sqrt{k}\ \sqrt{\Omega}\right]\right)$$

To reduce the size of the expressions in the analysis to follow, I introduce the parameter

```
def[β] = β == (-1)^(3/4) √k √Ω
```

$$\beta == (-1)^{3/4}\sqrt{k}\ \sqrt{\Omega}$$

```
w[17] = w[16] /. Sol[def[β], Ω] // PowerExpand // Expand // RE
```

$$V_k[R] == \frac{1}{\beta^2} - \frac{\text{BesselJ}[0,\ R\,\beta]}{\beta^2\,\text{BesselJ}[0,\ \beta]}$$

```
w[18] = w[17][[1]] == Collect[w[17][[2]], β]
```

$$V_k[R] == \frac{1 - \frac{\text{BesselJ}[0, R\,\beta]}{\text{BesselJ}[0, \beta]}}{\beta^2}$$

The solution will have a different character for large and small $\beta$

```
Module[{Vk},
 Vk[R_, β_] := Re[(1 - BesselJ[0,R β]/BesselJ[0,β])/β^2];
 Plot[{Vk[R, 0.1], Vk[R, 1], Vk[R, 10], Vk[R, 100], VPoiseuille[R]}, {R, 0, 1},
  PlotStyle → {Black, Blue, Green, Orange, Directive[RED, Dashed]},
  PlotLegends → {Stl["β = 1/10"], Stl["β = 1"], Stl["β = 10"],
    Stl["β = 100"], Stl["Poiseuille"]}, AxesLabel → {Stl["R"], Stl["Vk"]}]]
```




Note that for $\beta \lesssim 1$, the shape of the time-dependent flow is similar to that of Poiseuille. For $\beta >> 1$ the time-dependent flow has a quite different shape.

The analytical limiting form for $\beta << 1$ is easily obtained with

```
w[19] = Normal@Series[(1 - BesselJ[0,R β]/BesselJ[0,β])/β^2, {β, 0, 1}]
```

$$\frac{1}{4}\left(-1+R^2\right)$$

The $\beta >> 1$ limit requires a bit more care

```
w[20] = Normal@Series[(BesselJ[0, x]), {x, ∞, 0}]
```

$$\sqrt{\frac{2}{\pi}}\sqrt{\frac{1}{x}}\,\mathrm{Cos}\left[\frac{\pi}{4}-x\right]$$

```
w[21] = (1 - BesselJ[0,R β]/BesselJ[0,β])/β^2 /. BesselJ[0, a_] → Sqrt[2/π] Sqrt[1/a] Cos[π/4 - a] // PowerExpand
```

$$\frac{1-\frac{\text{Cos}\left[\frac{\pi}{4}-R\,\beta\right]\,\text{Sec}\left[\frac{\pi}{4}-\beta\right]}{\sqrt{R}}}{\beta^2}$$

```
Module[{∈k = 1},
 Plot[{Re[(1 - BesselJ[0,R β]/BesselJ[0,β])/β^2 /. β → 10], Re[(1 - Cos[π/4-R β] Sec[π/4-β]/Sqrt[R])/β^2 /. β → 10]},
  {R, 0, 1}, PlotStyle → {Black, Directive[Red, Dashed]},
  PlotLegends → {Stl["β = 10 exact"], Stl["β = 10 approx"]},
  AxesLabel → {Stl["R"], Stl["Vk"]}]]
```




The leading order approximation is quite good for $\beta = 10$.

Exercise TB 13-19 (c) calls for determining the character of the flow as a function of frequency.

```
w[18]
```

$$V_k[R] == \frac{1-\frac{\text{BesselJ}[0,R\,\beta]}{\text{BesselJ}[0,\beta]}}{\beta^2}$$

Consider the parameter $\beta$

```
def[β]
```

$$\beta == (-1)^{3/4}\sqrt{k}\,\sqrt{\Omega}$$

Then

```
w[22] = def[β] /. Sol[def[Ω], Ω] /. Sol[def[t0], t0]
```

$$\beta == (-1)^{3/4}\sqrt{k}\sqrt{\frac{a^2\,\omega_0}{\nu}}$$

The quantity W is known as the Womersley number

```
def[W] = W == Sqrt[a^2 ω0/ν]
```

$$W == \sqrt{\frac{a^2\,\omega_0}{\nu}}$$

From http://wwwf.imperial.ac.uk/~ajm8/BioFluids/lec1114.pdf, I find a representative value of W appropriate for blood flow.

> In a physiological context, $\alpha$ is known as the Womersley number, but similar parameters are well known in other contexts by other names.
>
> In the aorta, with $a = 0.015$, $2\pi/\omega = 1$ and $\nu = \mu/\rho = 4\times10^{-6}$ we have

```
Module[{R = 0.015 m, ν = 4×10^-6 (m^2 s), ω0 = (2 π) s},
 Sqrt[R^2 ω0/ν]]
```

18.7997

So a representative value is W = 19 and the $\beta$ >> 1 expansion is the relevant one for the present context of arterial flow. In more general contexts, this quantity is called the Strouhal number.

I use this information to consider a simplified form of $V_k$

```
w[23] = w[18] /. Sol[def[β], β] /. Sol[def[Ω], Ω]
```

$$V_k[R] == \frac{i\left(1-\frac{\text{BesselJ}\left[0,\frac{(1-i)\sqrt{k}\,R\sqrt{t0\,\omega_0}}{\sqrt{2}}\right]}{\text{BesselJ}\left[0,\frac{(1-i)\sqrt{k}\sqrt{t0\,\omega_0}}{\sqrt{2}}\right]}\right)}{k\,t0\,\omega_0}$$

As suggested in part (c), consider the frequency $\omega = k\,\omega_0$

```
w[24] = w[23] /. ω₀ → ω/k // PowerExpand
```

$$V_k[R] == \frac{i\left(1-\frac{\text{BesselJ}\left[0,\frac{(1-i)\,R\,\sqrt{t0}\,\sqrt{\omega}}{\sqrt{2}}\right]}{\text{BesselJ}\left[0,\frac{(1-i)\,\sqrt{t0}\,\sqrt{\omega}}{\sqrt{2}}\right]}\right)}{t0\,\omega}$$

Introduce the dimensionless parameter f = $\omega$ t0

```
w[25] = w[24] /. Sol[f == ω t0, t0] // PowerExpand
```

$$V_k[R] == \frac{i\left(1-\frac{\text{BesselJ}\left[0,\frac{(1-i)\,\sqrt{f}\,R}{\sqrt{2}}\right]}{\text{BesselJ}\left[0,\frac{(1-i)\,\sqrt{f}}{\sqrt{2}}\right]}\right)}{f}$$

```
Module[{ϵk = 1, ω0 = 1/(2π), F},
 F[x_, R_] :=
  Re[ i (1 - BesselJ[0, (1-i) √f R/√2] / BesselJ[0, (1-i) √f/√2]) ϵk / f ];
 Plot[F[f, 0.1], {f, 0.1, 20},
  PlotRange → All, AxesLabel → {Stl["f"], Stl["Vk[R=0.1]"]}]]
```




So — the answer to Exercise TB13-19 (3) is that the central flow is very similar to Poiseuille flow (order -0.25) for low frequency f << 1 and quite different for f >> 1.

With regard to Exercise TB13-19 part (d), the desired result just requires some manipulation of previous results.

$$V(R,\ T)\ = \sum_{k=0}^{n} \mathrm{Re}\left[V_n(R)\, e^{i\,k\,\Omega\,T}\right] = V_0 + \sum_{k=1}^{n} \epsilon_k\, \mathrm{Re}\left[V_n(R)\, e^{i\,k\,\Omega\,T}\right]$$

Recall

```
w[9] // RE
```

$$V_0[R] == \frac{1}{4}\left(-1 + R^2\right)$$

Recall the previous results

```
w[18]
```

$$V_k[R] == \frac{1 - \frac{\text{BesselJ}[0, R\,\beta]}{\text{BesselJ}[0, \beta]}}{\beta^2}$$

```
w[22]
```

$$\beta == (-1)^{3/4} \sqrt{k} \sqrt{\frac{a^2\, \omega_0}{\nu}}$$

```
def[W]
```

$$W == \sqrt{\frac{a^2\, \omega_0}{\nu}}$$

Then

```
w[26] = w[18] /. Sol[def[β], β] /. Sol[def[Ω], Ω] /. Sol[def[t0], t0] /.
   Sol[def[W], a] // PowerExpand
```

$$V_k[R] == \frac{i \left(1 - \frac{\text{BesselJ}\left[0, \frac{(1-i)\sqrt{k}\,R\,W}{\sqrt{2}}\right]}{\text{BesselJ}\left[0, \frac{(1-i)\sqrt{k}\,W}{\sqrt{2}}\right]}\right)}{k\, W^2}$$

The combined solution is

```
w[27]  =
```

$$\frac{1}{4}\left(-1+R^2\right) + \text{Inactive[Sum]}\left[\text{Re}\left[\frac{i\left(1-\frac{\text{BesselJ}\left[0,\frac{(1-i)\sqrt{k}\,R\,W}{\sqrt{2}}\right]}{\text{BesselJ}\left[0,\frac{(1-i)\sqrt{k}\,W}{\sqrt{2}}\right]}\right)\epsilon_k}{k\,W^2}\text{Exp}[I\,k\,\omega_0\,t]\right],\{k,1,n\}\right]$$

$$\frac{1}{4}\left(-1+R^2\right)+\sum_{k=1}^{n}-\text{Im}\left[\frac{e^{i\,k\,t\,\omega_0}\left(1-\frac{\text{BesselJ}\left[0,\frac{(1-i)\sqrt{k}\,R\,W}{\sqrt{2}}\right]}{\text{BesselJ}\left[0,\frac{(1-i)\sqrt{k}\,W}{\sqrt{2}}\right]}\right)\epsilon_k}{k\,W^2}\right]$$

which is equivalent to the result TB 13.85

For part (e) I construct

```
Clear[Vk, Pk];
Vk[R_, t_, k_, ω0_, W_, ϵk_] :=
  Re[ (ⅈ (1 - BesselJ[0, (1-ⅈ) √k R W/√2] / BesselJ[0, (1-ⅈ) √k W/√2]) ϵk) / (k W^2) Exp[I k ω0 t]];
Pk[t_, k_, ω0_] := Re[Exp[I k ω0 t]]
```

When the Womersley number W << 1 (diffusion time is much longer than oscillation time), there is balance between the driving term and the viscous force. The velocity is in phase with the driving term.

```
Module[{k = 1, ω0 = 1/(2π), R = 0.1, W = 0.1, ϵk = 10},
 Plot[{Pk[t, k, ω0], Vk[R, t, k, ω0, W, ϵk]},
  {t, 0, 50}, PlotStyle → {Black, Blue}, PlotLegends →
   {Stl["pressure driver"], Stl["fluid response"]}, AxesLabel → {Stl["t"], ""}]]
```




When the Womersley number W >>1 (diffusion time is much slower than oscillation time), there is balance between the driving term and the inertial term. The velocity is out of phase with the driving term.

```
Module[{k = 1, ω0 = 1/(2π), R = 0.1, W = 5, ϵk = 10},
 Plot[{Pk[t, k, ω0], Vk[R, t, k, ω0, W, ϵk]}, {t, 0, 50}, PlotStyle → {Black, Blue},
  PlotLegends → {Stl["pressure driver"], Stl["fluid response"]},
  AxesLabel → {Stl["t"], ""}]]
```




I consider the boundary layer that forms near the edge of an arterey when W >> 1

```
Module[{k = 1, ω0 = 1/(2π), R = 0.1, W = 10, ϵk = 10},
 Plot[{Vk[R, 0, k, ω0, W, ϵk], Vk[R, 2.5, k, ω0, W, ϵk],
   Vk[R, 5, k, ω0, W, ϵk], Vk[R, 10, k, ω0, W, ϵk], 1/4 (-1 + R^2)}, {R, 0, 1},
  AxesLabel → {Stl["R"], ""}, PlotStyle →
   {Black, Blue, Green, Orange, Directive[Black, Dashed]}, PlotLegends →
   {Stl["T = 0"], Stl["T = 2.5"], Stl["T = 5"], Stl["T = 10"], Stl["Poiseuille"]}]]
```




Consider the T = 0 flow.

```
w[26] = Vk[R, 0, k, ω0, W, ϵk] /. k → 1
```

$$-\mathrm{Im}\left[\frac{\epsilon k\left(1-\frac{\mathrm{BesselJ}\left[0,\frac{(1-i)\,R\,W}{\sqrt{2}}\right]}{\mathrm{BesselJ}\left[0,\frac{(1-i)\,W}{\sqrt{2}}\right]}\right)}{W^2}\right]$$

and focus on radial dependence

$$\texttt{w[27]} = \left(1-\frac{\mathrm{BesselJ}\left[0,\frac{(1-i)\,R\,W}{\sqrt{2}}\right]}{\mathrm{BesselJ}\left[0,\frac{(1-i)\,W}{\sqrt{2}}\right]}\right)$$

$$1-\frac{\mathrm{BesselJ}\left[0,\frac{(1-i)\,R\,W}{\sqrt{2}}\right]}{\mathrm{BesselJ}\left[0,\frac{(1-i)\,W}{\sqrt{2}}\right]}$$

```
Module[{F},
 F[R_, W_] := -Im[(1 - BesselJ[0, ((1-i) R W)/√2]/BesselJ[0, ((1-i) W)/√2])];
 Plot[{F[R, 1], F[R, 10], F[R, 20], F[R, 50]}, {R, 0, 1},
  AxesLabel → {Stl["R"], ""}, PlotStyle → {Black, Blue, Green, Orange},
  PlotLegends → {Stl["W = 5"], Stl["W = 10"], Stl["W = 20"], Stl["W = 50"]},
  PlotLabel → Stl["Boundary layer becomes pronounced at high W"]]]
```




I have not previously considered analytical solutions of boundary layers in fluids so I will content myself with a geometrical/numerical answer to TB13-19 part (e). Later, I intend to return and obtain an analytical result.

I define a function to estimate the width of the boundary layer

```
Clear[FRadial, EstimateBoundaryLayerThickness];
FRadial[R_, W_] := -Im[(1 - BesselJ[0, (1-ⅈ) R W/√2] / BesselJ[0, (1-ⅈ) W/√2])];

(* I estimate the thickness of the boundary layer to be the half-
 width of the negative pulse near the edge of the artery. *)
EstimateBoundaryLayerThickness[W_, RMinGuess_] :=
 Module[{result, min, RVal, RHalfInner, RHalfOuter, δBoundaryLayer, F},
  (* find minimum near R = 1 *)
  result = FindMinimum[{FRadial[R, W], 0.7 < R < 0.999}, {R, 0.95}];
  {min, RVal} = {result [[1]], result [[2, 1, 2]]} ;
  (* find the Rinner value at which F = min/2 *)
  RHalfInner = FindRoot[FRadial[R, W] == min / 2, {R, RVal - 0.02}][[1, 2]] ;
  (* find the Router value at which F = min/2 *)
  RHalfOuter = FindRoot[FRadial[R, W] == min / 2, {R, RVal + 0.02}][[1, 2]] ;
  (* estimate δ as Router - Rinner *)
  δBoundaryLayer = RHalfOuter - RHalfInner]
```

I calculate $\delta$(W) numerically and then obtain a functional fit

```
Module[{values, fit, valuesFit, aFit, bFit, lab},
 values = Table[{W, EstimateBoundaryLayerThickness[W, 0.95]}, {W, 10, 50, 5}];
 fit = FindFit[values, a W^b, {a, b}, W];
 valuesFit = Interpolation@Table[{W, a W^b} /. fit, {W, 10, 50, 5}];
 lab = Stl@StringForm["fitting function = a x^b   a = ``, b = ``", a, b] /. fit;
 Plot[valuesFit[W], {W, 10, 50}, AxesLabel → {Stl["W"], Stl["δ"]},
  Epilog → {Red, PointSize[0.02], Point /@ values}, PlotLabel → lab]]
```




The numerical result strongly suggest $\delta \sim 1/W$.

# References

On pulsatile flow
https://www.youtube.com/watch?v=HFUSnqaFio0
https://www.youtube.com/watch?v=b029wRQnZwI&list=PLbMVogVj5nJRsXU2WZyMTIDq6tPCjlQB3&index=17

nice summary of solving inhomogeneous PDE using eigenfunction expansions
https://www.youtube.com/watch?v=HoLweaiYG7g

Detailed analysis of Womersley flow
https://www.google.com/search?q=womersley+flow&oq=womersley&aqs=chrome.1.69i57j0l5.5363j0j8&sourceid=chrome&ie=UTF-8

Wikipedia has a nice treatment of pulsatile flow
https://en.wikipedia.org/wiki/Pulsatile_flow

Mathematica demonstration
http://demonstrations.wolfram.com/PulsatileFlowInACircularTube/

Nice physical pictures in the context of blood flow
http://www.mate.tue.nl/people/vosse/docs/cardio.pdf